
1. [bookmark: _Toc122434485][bookmark: _Toc141170366]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D23wk24 related to the title of this CR.
[bookmark: _Hlk90298041]Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc141170367]Corrections required

1. [bookmark: _Toc136279094][bookmark: _Toc141170368]Change 1
	Function name
	fl_TC_11_4_1_TestBody

	Reason for change
	1. The v_GSM_MobilityInfo could be rewritten in step 3Ca1 in the function f_Check_NG_PDUSessionReleaseReq which would effectivelly set the PDNindex to omit and hence the information would be lost and cause runtime error later when calling f_IP_RtpRtcpLoopback with omitted value.
2. In steps 3Fa / 3Fb the old timer is still referenced.
3. At step 3E, UE can send deregister request immediately after sending PDU session release complete to start normal service once Emergency call end’s. f_NR5GC_PDUSessionRelease function may time to complete its all subfunction’s like f_IP_Handling_StopPDN to stop all DHCP and ICMPv6 servers.
4. Since the f_NR_RRCSetup_Def (called by f_NR5GC_RRC_Idle_Steps3_20) will not start the UL grants and send the RRC message for non-early contention tests it has to be added in step 9.

	Summary of change
	1. Stored PDNindex separately and used its value when calling f_IP_RtpRtcpLoopback.
2. In steps 3Fa / 3Fb corrected the typo in timer name.
3. Added default variable to handle deregistration request as Note 2 before function f_NR5GC_PDUSessionRelease. And removed corresponding function of Note 2 from alt statement after step3F.
4. Added start of UL grant and sending of RRC setup in step 9.

	TTCN module
	11_4\Emergency_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	  function fl_TC_11_4_1_TestBody() runs on NR5GC_PTC
  {
…

    //@siclog "Step 3" siclog@
    //Make the UE release the emergency call.
    //@siclog "Step 3A" siclog@
    //The Generic test procedure for IMS MO Emergency call release
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2], cms_IMS_IPCAN_Trigger("Trigger MO Call Release")); // @sic R5-233779 sic@
    // Don't need this anymore as next step is no longer a DL message    f_Delay(0.1);//@sic R5s230143 sic@ :ensure that the Ack for PDU Session Release Complete is sent in DL.

    //@sic R5-233779 sic@:Step3B-Step3F
    // Get this for the emergency PDN now in case UE doesn't send a PDU Release Req
    v_GSM_MobilityInfo_IMS := f_Get_PDUSessionForDNNType(v_PDUSessionInfo, Emergency_PDN);
    v_GSM_MobilityInfo := valueof(v_GSM_MobilityInfo_IMS);


    v_PDUSessionId := v_GSM_MobilityInfo.SessionId;
    
    //@siclog "Step 3B" siclog@
    //SS starts timer T1 = 5 seconds.
    t_Wait.start;
    //EXCEPTION: Steps 3Ca1 to 3Cb1 describe a transaction that depends on the UE behaviour; the "lower case letter" identifies a step sequence that takes place if a specific behaviour happens.
    alt {
      [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Trigger ("IMS Call Release Msgs Sent"))
       { repeat}

      //@siclog "Step 3Ca1" siclog@
      //The UE transmits a PDU SESSION RELEASE REQUEST message.
      [] SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
       {
         t_Wait.stop; // Step 3Ca2
         if (not f_Check_NG_PDUSessionReleaseReq (v_GSM_MobilityInfo, v_PDUSessionInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionId))
          {
             f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Request failed");
          }
       }
      //@siclog "Step 3Cb1" siclog@
      //Timer T1 expires.
      [] t_Wait.timeout
       {
          //Do nothing
       }
    }
    //@siclog "Step 3D" siclog@
    //The SS transmits a PDU SESSION RELEASE COMMAND message.
    //@siclog "Step 3E" siclog@
    //Check: Does the UE transmit a PDU SESSION RELEASE COMPLETE message?
    v_PDUSessionId := v_GSM_MobilityInfo.SessionId;
    f_NR5GC_PDUSessionRelease(nr_Cell1, v_PDUSessionId);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 3E");
    //@siclog "Step 3Fa1-3Fb3" siclog@
    t_Wait2.start;
    //EXCEPTION: EXCEPTION: Steps 3Fa1-3Fb3 describe optional behaviour that depends on the UE implementation. the "lower case letter" identifies a step sequence that take place if a UE specific behavior takes place.
    alt {
      //@siclog "Step 3Fa2" siclog@
      //Check: Does the UE transmit RRCSetupRequest on NR Cell 1?
      [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest(?, ?)))
       {
         t_Wait.stop;
         f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 3Fa2");
         //@siclog "Step 3Fa3" siclog@
         //SS locally releases the RRC connection.
         f_NR_RRCRelease_Local(nr_Cell1, f_NR_GetNextSendOccasion_DL(nr_Cell1))
       }
      //@siclog "Step 3Fb1" siclog@
      //Timer T2 expires
      [] t_Wait.timeout
       {
          //@siclog "Step 3Fb2" siclog@
          //The SS transmits an RRCRelease message
          f_NR_RRCRelease(nr_Cell1);
          //@siclog "Step 3Fb3" siclog@
          //Check: Does the UE transmit RRCSetupRequest on NR Cell 1?
          SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest(?, ?)));
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 3Fb2");
       }

          //@siclog "NOTE 2" siclog@
      [] SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1,
                                           tsc_NR_RbId_SRB2,
                                           cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,
                                                         cr_NG_DEREGISTRATION_REQUEST_MO(crs_DeregisterType('0'B, '0'B,'01'B),?,?)))) -> value v_ReceivedAsp
        {
          repeat;
        }
    }
    // PDU session definitely gone now, so can stop can stop looping back RTP/RTCP
    f_IP_RtpRtcpLoopback(IP, v_GSM_MobilityInfo.PdnIndex, tsc_IMS_MediaPort_M1, omit);
    
    //@sic Rs230583 sic@
    //@siclog "Step 4A-4B" siclog@
    //Void
    //@sic R5-233779 sic@
    //@siclog "Step 9A1-9A2" siclog@
    //Void
    //@siclog "Step 9A3 - 9A21a1" siclog@
    //Steps 2-20a1 of Table 4.5.2.2- 2: Registration procedure for initial registration as specified in TS 38.508-1 [4] are performed on NR Cell 1.

    f_NR5GC_RRC_Idle_Steps3_20(nr_Cell1, TESTMode_OFF, -, Initial_NoSecurity);
    //@siclog "Step 9A22" siclog@






After Change:
	  function fl_TC_11_4_1_TestBody() runs on NR5GC_PTC
  {
…

    var PDN_Index_Type v_PDNindex;
var default v_MyDefaultVar := null;

    //@siclog "Step 3" siclog@
    //Make the UE release the emergency call.
    //@siclog "Step 3A" siclog@
    //The Generic test procedure for IMS MO Emergency call release
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2], cms_IMS_IPCAN_Trigger("Trigger MO Call Release")); // @sic R5-233779 sic@
    // Don't need this anymore as next step is no longer a DL message    f_Delay(0.1);//@sic R5s230143 sic@ :ensure that the Ack for PDU Session Release Complete is sent in DL.

    //@sic R5-233779 sic@:Step3B-Step3F
    // Get this for the emergency PDN now in case UE doesn't send a PDU Release Req
    v_GSM_MobilityInfo_IMS := f_Get_PDUSessionForDNNType(v_PDUSessionInfo, Emergency_PDN);
    v_GSM_MobilityInfo := valueof(v_GSM_MobilityInfo_IMS);
    v_PDNindex := v_GSM_MobilityInfo.PdnIndex;
    v_PDUSessionId := v_GSM_MobilityInfo.SessionId;
    
    //@siclog "Step 3B" siclog@
    //SS starts timer T1 = 5 seconds.
    t_Wait.start;
    //EXCEPTION: Steps 3Ca1 to 3Cb1 describe a transaction that depends on the UE behaviour; the "lower case letter" identifies a step sequence that takes place if a specific behaviour happens.
    alt {
      [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Trigger ("IMS Call Release Msgs Sent"))
       { repeat}

      //@siclog "Step 3Ca1" siclog@
      //The UE transmits a PDU SESSION RELEASE REQUEST message.
      [] SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
       {
         t_Wait.stop; // Step 3Ca2
         if (not f_Check_NG_PDUSessionReleaseReq (v_GSM_MobilityInfo, v_PDUSessionInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionId))
          {
             f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Request failed");
          }
       }
      //@siclog "Step 3Cb1" siclog@
      //Timer T1 expires.
      [] t_Wait.timeout
       {
          //Do nothing
       }
    }
    //@siclog "Step 3D" siclog@
    //The SS transmits a PDU SESSION RELEASE COMMAND message.
    //@siclog "Step 3E" siclog@
    //Check: Does the UE transmit a PDU SESSION RELEASE COMPLETE message?
    v_PDUSessionId := v_GSM_MobilityInfo.SessionId;
    //@siclog "NOTE 2" siclog@
    v_MyDefaultVar := activate(a_NR5GC_Deregistration_REQ(nr_Cell1));
    f_NR5GC_PDUSessionRelease(nr_Cell1, v_PDUSessionId);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 3E");
    //@siclog "Step 3Fa1-3Fb3" siclog@
    t_Wait2.start;
    //EXCEPTION: EXCEPTION: Steps 3Fa1-3Fb3 describe optional behaviour that depends on the UE implementation. the "lower case letter" identifies a step sequence that take place if a UE specific behavior takes place.
    alt {
      //@siclog "Step 3Fa2" siclog@
      //Check: Does the UE transmit RRCSetupRequest on NR Cell 1?
      [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest(?, ?)))
       {
         t_Wait2.stop; 
         f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 3Fa2");
         //@siclog "Step 3Fa3" siclog@
         //SS locally releases the RRC connection.
         f_NR_RRCRelease_Local(nr_Cell1, f_NR_GetNextSendOccasion_DL(nr_Cell1))
       }
      //@siclog "Step 3Fb1" siclog@
      //Timer T2 expires
      [] t_Wait2.timeout
       {
          //@siclog "Step 3Fb2" siclog@
          //The SS transmits an RRCRelease message
          f_NR_RRCRelease(nr_Cell1);
          //@siclog "Step 3Fb3" siclog@
          //Check: Does the UE transmit RRCSetupRequest on NR Cell 1?
          SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest(?, ?)));
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 3Fb2");
       }
    }
deactivate(v_MyDefaultVar);
    // PDU session definitely gone now, so can stop can stop looping back RTP/RTCP 
    f_IP_RtpRtcpLoopback(IP, v_PDNindex, tsc_IMS_MediaPort_M1, omit);
    
    //@sic Rs230583 sic@
    //@siclog "Step 4A-4B" siclog@
    //Void
    //@sic R5-233779 sic@
    //@siclog "Step 9A1-9A2" siclog@
    //Void
    //@siclog "Step 9A3 - 9A21a1" siclog@
    //Steps 2-20a1 of Table 4.5.2.2- 2: Registration procedure for initial registration as specified in TS 38.508-1 [4] are performed on NR Cell 1.
    f_NR_ULGrantConfiguration_Start(nr_Cell1);
    SRB.send(cas_NR_SRB0_RrcPdu_REQ(nr_Cell1,
                                      cs_TimingInfo_Now,
                                      f_NR_38508_RRCSetup(nr_Cell1, tsc_NR_RRC_TI_Def)));

    f_NR5GC_RRC_Idle_Steps3_20(nr_Cell1, TESTMode_OFF, -, Initial_NoSecurity);
    //@siclog "Step 9A22" siclog@





